
  _________________________________________

  Titolo: Ordinamento esterno

  Data: 01/10/2009

  Programmatore: Valerio Finizio

  Obiettivo dell'esercitazione:

    L'allievo è in grado di scrivere il

    programma per ordinare un file.

  Strategia risolutiva:

    Suddivisione del file in sottoarchivi di dimensione

    tale da poter essere ordinati in memoria centrale.

    Fusione in memoria di massa dei sottoarchivi.

  _________________________________________

  Compilatore: Dev-Pascal 1.9.2

  Sistema Operativo: Windows XP

}

program es3;

uses crt;

const

  dim=5;

  numero_record=24;

type

  rectype=record

            chiave:integer;

            info:char;

          end;

var

  f:file of rectype;

  nff:string; (*nome fisico del file*)

  app,rec:rectype;

  i,j,numero_archivi:integer;

  num_file,nome_file:string;

procedure creafile(nffpar:string);

var

  i,num:integer;

begin

  assign(f,nffpar);

  rewrite(f);

  (* I record vengono caricati automaticamente

     generando casualmente caratteri da memorizzare

     nel campo info e numeri compresi tra 0 e 99 nel

     campo chiave *)

  randomize;

  for i:= 1 to numero_record do

    begin

      app.chiave:=random(100);

      write (app.chiave,' ');

      num:=random(100);

      app.info:=chr(num);

      write(f,app);

    end;

  writeln;

  close(f);

end;

procedure ordina(nffpar:string);

(* Obiettivo procedura: Creare i sottoarchivi e ordinarli *)

var

  vet:array[1..dim] of rectype; (* Vettore di appoggio per l'ordinamento

                                   dei sottoarchivi*)

  i,j,k,h,cont:byte;

  app:rectype;

  nome_file,num_file:string;

  record_rimasti,numero_archivi:integer;

  f1:file of rectype;

begin

  numero_archivi:=(numero_record div dim);

  assign(f,nffpar);

  reset(f);

  (* Ciclo di creazione dei vari sottoarchivi*)

  for i :=1 to numero_archivi do

    begin

    (* Ciclo di trasferimento dei records di un sottoarchivio alla volta

       nel vettore per procedere al suo ordinamento *)

    for j := 1 to dim do read(f,vet[j]);     (*trasferimento di un sottoarchivio

                                               nel vettore*)

    (*Cicli annidati per l'ordinamento del sottoarchivio in memoria*)

    for k:=1 to (dim-1) do

      for h:=(k+1) to dim do

        if (vet[k].chiave>vet[h].chiave) then

          begin

            app:=vet[k];

            vet[k]:=vet[h];

            vet[h]:=app;

          end;

     for cont:=1 to dim do write(vet[cont].chiave,' '); (*controllo a video*)

     writeln;

    (*fine dell'ordinamento di un sottoarchivio *)

    (*Salvataggio del sottoarchivio ordinato in un file

      opportunamente denominato*)

    str(i,num_file); (*A num_file viene assegnato il carattere corrispondente

                       al valore numerico di i*)

    nome_file:=concat('s',num_file); (* Il nome del sottoarchivio è la

                                        concatenazione di s con il carattere

                                        contenuto in num_file*)

    assign(f1,nome_file);

    rewrite(f1);

    for j := 1 to dim do write(f1,vet[j]);

    close(f1);

    (*Fine del salvataggio di un sottoarchivio*)

  end;

  (*Fine creazione e ordinamento dei sottoarchivi completi, cioè

    con dimensione uguale a "dim" *)

  (*Eventuale creazione dell'ultimo sottoarchivio con un numero di record

    inferiore a "dim",nel caso che il numero di record dell'intero file

    da ordinare diviso il numero di record di un sottoarchivio presenti il

    resto diverso da 0 *)

  record_rimasti:=(numero_record-(dim*numero_archivi));

    for j:=1 to record_rimasti do read(f,vet[j]);  (*trasferimento nel vettore *)

    (*Cicli annidati per l'ordinamento*)

      for k:=1 to (record_rimasti-1) do

        for h:=(k+1) to record_rimasti do

          if (vet[k].chiave>vet[h].chiave) then

            begin

              app:=vet[k];

              vet[k]:=vet[h];

              vet[h]:=app;

            end;

      for cont:=1 to record_rimasti do write(vet[cont].chiave,' ');

  i:=i+1;

  str(i,num_file);

  nome_file:=concat('s',num_file);

  assign(f1,nome_file);

  rewrite(f1);

  for j := 1 to record_rimasti do write(f1,vet[j]); (*Controllo a video

                                                      dell'ultimo sottoarchivio*)

  close(f1);

  (*Fine della creazione e ordinamento dell'ultimo sottoarchivio*)

  close(f);

end;

procedure ordina_archivio(nffpar:string);

var

  f,f1:file of rectype;

  i:integer;

  rec:rectype;

begin

  (*Apertura in creazione del file da ordinare*)

  assign(f,nffpar);

  rewrite(f);

  (*Apertura in lettura del primo sottoarchivio*)

  assign(f1,'s1');

  reset(f1);

  (*Trasferimento di f1(sottoarchivio s1) in f(file originale da ordinare)*)

  for i:=1 to dim do

    begin

      read(f1,rec);

      write(f,rec);

    end;

  close(f);

  close(f1);

  erase(f1);

end;

procedure fusione(nffpar:string;sottoarcpar:string);

(* Obiettivo: Fondere i sottoarchivi per riscrivere il file originale in modo

              ordinato rispetto alla chiave

   Strategia risolutiva: Dopo aver ricreato il file originale memorizzando in

                         esso il primo sottoarchivio, si procede con la fusione

                         del file originale con il secondo sottoarchivio, poi

                         con la fusione del file originale con il terzo e così

                         via fino all'ultimo sottoarchivio

*)

var

  f,sf,ord:file of rectype;

  rec1,rec2:rectype;

begin

  (*Apertura in lettura del file originale*)

  assign(f,nffpar);

  reset(f);

  (*Apertura in lettura dei sottoarchivi: dal secondo in poi*)

  assign(sf,sottoarcpar);

  reset(sf);

  (*Apertura in scrittura del file di appoggio per la fusione*)

  assign(ord,'ordinato');

  rewrite(ord);

  read(f,rec1);

  read(sf,rec2);

  (*Ciclo per la fusione di f e di sf in ord *)

  while filesize(ord)< (filesize(f)+filesize(sf)) do

    if rec1.chiave<rec2.chiave then

      begin

        write (ord,rec1);

        if not EOF(f) then read(f, rec1)

                      else begin

                             write(ord,rec2);

                             while not EOF(sf) do

                               begin

                                 read(sf,rec2);

                                 write (ord,rec2);

                               end;

                           end;

      end

  else

    begin

      write(ord,rec2);

      if not EOF(sf) then read (sf,rec2)

                     else begin

                            write (ord,rec1);

                            while not EOF(f) do

                              begin

                                read (f,rec1);

                               write (ord,rec1);

                              end;

                     end;

    end;

  (* Copia di ord in f*)

  rewrite(f);

  reset(ord);

  while not EOF(ord) do

    begin

      read(ord,rec1);

      write(f,rec1);

    end;

  close(f);

  close(sf);

  close(ord);

  (*Cancellazione del file ord e del sottoarchivio oggetto della fusione*)

  erase(ord);

  erase(sf);

end;

BEGIN

  clrscr;

  nff:='subarc.dat';

  creafile(nff);

  readln;

  ordina(nff);

  readln;

  writeln;

  writeln;

  ordina_archivio(nff);

  (*Calcolo del numero dei sottoarchivi da creare*)

  numero_archivi:=(numero_record div dim);

  if (numero_archivi*dim)<numero_record then numero_archivi:=numero_archivi+1;

  (*Ciclo che effettua la fusione dei sottoarchivi*)

  for i:=2 to (numero_archivi) do

  begin

    str(i,num_file);

    nome_file:=concat('s',num_file);

    fusione (nff,nome_file);

  end;

  assign(f,nff);

  reset(f);

  for i:=1 to numero_record do

    begin

      read(f,rec);

      write(rec.chiave,' ')(*Controllo a video de risultato*)

    end;

  readln

END.

